
1) DESCRIÇÃO:


Este trabalho possui duas partes. Na primeira, o objetivo é implementar a semântica operacional estrutural (em inglês, structural operational semantics - SOS) em PROLOG para um sub-conjunto desta mesma linguagem, com:

· predicados sem variáveis;

· corte;

· falha, e

· supondo uma semântica fixa e arbitrária para certos predicados.

Já na segunda parte, o objetivo é fazer o SOS para o comando goto, comum em várias linguagens de programação, de modo elegante.

2) SOS DE UM SUB-CONJUNTO DE PROLOG:

2.1) BNF DA LINGUAGEM:


Para poder definir a semântica de uma linguagem, primeiro é necessário definir a BNF de sua sintaxe, que neste caso, é a seguinte:

< programa > ::= < objetivo >

| < sentenças > < objetivo >

< sentenças > ::= < sentença >




 | < sentença > < sentenças >

< sentença > ::= < predicado > “.”
               | < predicado > “:-“ < resto > “.”

< objetivo > ::= “:-“ < resto > “.”

< resto > ::= < unidade >

  | < unidade > “,” < resto >
< unidade > ::= < predicado >



    | “fail”



    | “!”



    | “read(“ < variável > ”)”



    | “write(” < variável > “)”

< predicado > ::= < letra minúscula >




 | < letra minúscula > < caracteres >

< variável > ::= < letra maiúscula >




| < letra maiúscula > < caracteres >

< letra minúscula > ::= “a” | ... | “z”

< letra maiúscula > ::= “A” | ... | “Z”

< número > ::= “0” | ... | “9”

< caracteres > ::= < caractere > 

       | < caractere > < caracteres >

< caractere > ::= < letra minúscula > 

      | < letra maiúscula >

      | < número >

      | “_” 

2.2) SOS da Linguagem:

Com a BNF da linguagem, pode-se definir as regras de inferência para a SOS da mesma. A estrutura usada é a seguinte:

< :- O ., LC, E, S, M, A >, onde:

· O é o objetivo;

· LC é a lista de cláusulas dada no programa;

· E é a lista de entradas;

· S é a lista de saídas;

· M é o último endereço de memória utilizado, e

· A é o ambiente, o qual é uma pilha.

Já as funções auxiliares são as definidas abaixo:

· mud( O, M ): muda o nome das variáveis que aparece no objetivo O nos read’s e write’s para serem endereços de memória, únicos em todo o programa, de forma que os endereços começam em M e vão até                          M + vars( O ) - 1 em O;

· vars( O ): devolve o número de váriáveis distintas do objetivo O;

· cortar( A, LC, O ): retira do ambiente A as cláusulas que são cortadas com o corte, que são os pontos de escolha abertos na cláusula onde ele aparece e os da cláusula onde ele aparece. O objetivo atual permite se saber se houve pontos de escolha abertos antes do corte, pois O aparecerá no rabo das primeiras cláusulas se isto acontecer. E LC permite saber se houve outras opções para a sentença que tinha o corte, pois depois de remover as do primeiro caso (se existirem), verifica se as próximas são alternativas para a que continha o corte e as retira de A se for verdade;

· LC_1( p, C ): pega a primeira cláusula de p dada em LC e a expande, concatenando-a com a C. se p for fato, retorna apenas C. Se C for true, retorna apenas a primeira cláusula de p. Se p for fato e C, true, retorna true;
· LC_R( p, C ): pega todas as cláusulas de p com exceção da primeira dada em LC e as expande, concatenando-as com a C e retornando a lista de cláusulas resultantes. se p for fato, retorna apenas C na lista. Se C for true, retorna apenas as cláusulas de p com a exceção da primeira. Se p for fato e C, true, coloca true na lista. Se só tiver uma cláusula de p em LC, a lista retornada é vazia (não há pontos de escolha), e
· trans( O, X, VE ): troca todas as ocorrências da variável X que aparecem em O pelo seu valor VE.
A seguir, encontram-se as regras de SOS:

< :- true ., LC, E, S, M, A > ( < :- true ., LC, E, S, M, A >

< :- fail ., LC, E, S, M, [ ] > ( < :- fail ., LC, E, S, M, [ ] >
< :- fail, R ., LC, E, S, M, [ ] > ( < :- fail ., LC, E, S, M, [ ] >

< :- fail ., LC, E, S, M, [ TA | RA ] > ( < :- TA’ ., LC, E, S, M’, RA > 

onde TA’ = mud( TA, M ) e M’ = M + vars( TA’ )
< :- fail, R ., LC, E, S, M, [ TA | RA ] > ( < :- TA’ ., LC, E, S, M’, RA >

onde TA’ = mud( TA, M ) e M’ = M + vars( TA’ )
< :- ! ., LC, E, S, M, A > ( < :- true ., LC, E, S, M, cortar( A, LC, true ) >

< :- !, R ., LC, E, S, M, A > ( < :- R ., LC, E, S, M, cortar( A, LC, R ) >

< :- p ., LC, E, S, M, A > ( < :- LC_1( p, true )’ ., LC, E, S, M’, [ LC_R( p, true ) | A ] >

onde LC_1( p, true )’ = mud( LC_1( p, true ), M ) e M’ = M + vars( LC_1( p, true )’ )
< :- p, R ., LC, E, S, M, A > ( < :- LCI_1( p, R )’ ., LC, E, S, M’, [ LC_R( p, R ) | A ] >

onde LC_1( p, R )’ = mud( LC_1( p, R ), M ) e M’ = M + vars( LC_1( p, R )’ )

< :- read( X ) ., LC, [ TE | RE ], S, M, A > ( < :- true ., LC, RE, S, M, A > 

< :- read( X ), R ., LC, [ TE | RE ], S, M, A > ( < :- R’ ., LC, RE, S, M, A >

onde R’ = trans( R, X, VE )

< :- read( V ) ., LC, [ V | RE ], S, M, A > ( < :- true ., LC, RE, S, M, A > 

< :- read( V ), R ., LC, [ V | RE ], S, M, A > ( < :- R ., LC, RE, S, M, A >

< :- read( V ) ., LC, [ TE | RE ], S, M, A > ( < :- fail ., LC, RE, S, M, A >

onde V ( TE

 < :- read( V ), R ., LC, [ TE | RE ], S, M, A > ( < :- fail ., LC, RE, S, M, A >

onde V ( TE

< :- write( X ) ., LC, E, S, M, A > ( < :- true ., LC, E, [ _ | S ], M, A >

< :- write( X ), p ., LC, E, S, M, A > ( < :- p ., LC, E, [ _ | S ], M, A >

< :- write( V ) ., LC, E, S, M, A > ( < :- true ., LC, E, [ V | S ], M, A >

< :- write( V ), p ., LC, E, S, M, A > ( < :- p ., LC, E, [ V | S ], M, A >

2.3) Implementação da SOS da Linguagem:


A implementação foi dividida em 4 módulos. Os testes feitos com programas usando esta linguagem comprovaram o funcionamento da implementação do SOS. A seguir, o código fonte de cada um deles:

:- [ 'lexica.pl' ]    ,

   [ 'sintatica.pl' ] ,

   [ 'semantica.pl' ] .

prog_principal                         :-

   fazerAnaLexica( Tokens )            ,

   !                                   ,

   fazerAnaSintatica( Tokens, Arvore ) ,

   !                                   ,

   fazerSOS( Arvore )                  ,

   !                                                        

serEspaco( 32 ) .     

serTab( 9 ) .          

serFimLinha( 10 ) .   

serFimArq( 26 ) .   

serFimArq( -1 ) .

serReserv( fail ) .

serSimples( 33, corte )   .   % !

serSimples( 40, parene )  .   % ( 

serSimples( 41, parend )  .   % )

serSimples( 44, virgula ) .   % ,  

serSimples( 46, ponto )   .   % .

serDuplo( 58, doisPontos ) .  % :  

serPar( 58, 45, se ) .        % :-       

serBranco( Car )      :- 

   serEspaco( Car )   ; 

   serTab( Car )      ; 

   serFimLinha( Car ) .

serMinuscula( Car ) :-        % a, ..., z

   Car >= 97        ,

   Car =< 122       .       

serMaiuscula( Car ) :-        % A, ..., Z 

   Car >= 65        ,

   Car =< 90        .       

serNumero( Car ) :-           % 0, ..., 9

   Car >= 48     ,

   Car =< 57     .

serCarVal( Car )       :-     % a, ..., z, A, ..., Z, 0, ..., 9, _

   serMinuscula( Car ) ;

   serMaiuscula( Car ) ;

   serNumero( Car )    ;

   Car == 95           .

fazerAnaLexica( Tokens )                                                   :- 

   write( '\nEscaneando: Análise léxica...\n' )                            , 

   write( 'Entre nome do arquivo fonte entre \' \' e seguido de ponto: ' ) ,  

   read( Arquivo )                                                         , 

   see( Arquivo )                                                          ,

   nl                                                                      , 

   escanear( Tokens )                                                      ,  

   seen                                                                    , 

   write( '\nAnálise léxica feita com sucesso.\n' )                        , 

   write( '\ntokens = ' )                                                  ,

   write( Tokens )                                                         , 

   nl                                                                      .

escanear( [ Token | ListaTokens ] )         :-

   tab( 4 )                                 , 

   pegarCar( Car )                          , 

   pegarToken( Car, Token, NovoCar )        , 

   restoProg( Token, NovoCar, ListaTokens ) , 

   !                                        .

pegarCar( Car )         :- 

   get0( Car )          , 

   ( serFimLinha( Car ) , 

     nl                 ,

     tab( 4 )           ; 

     serFimArq( Car )   , 

     nl                 ; 

     put( Car ) )       .

restoProg( eop, _, [ ] ) .

restoProg( _, Car, [ Token | ListaTokens ] ) :- 

   pegarToken( Car, Token, NovoCar )         , 

   restoProg( Token, NovoCar, ListaTokens )  .

pegarToken( Car, eop, 0 ) :-  % lê fim de arquivo 

   serFimArq( Car )       .

pegarToken( Car, Token, NovoCar ) :-  % lê token com 1 caractere 

   serSimples( Car, Token )       , 

   pegarCar( NovoCar )            .

pegarToken( Car, Token, NovoCar_2 )  :-  % lê token com 2 caracteres

   serDuplo( Car, OutroToken )       ,

   pegarCar( NovoCar_1 )             ,

   ( serPar( Car, NovoCar_1, Token ) ,

     pegarCar( NovoCar_2 )           ; 

     Token = OutroToken              ,

     NovoCar_2 = NovoCar_1 )         .

pegarToken( Car, Token, NovoCar_2 )          :-  % lê predicado ou fail

   serMinuscula( Car )                       , 

   pegarCar( NovoCar_1 )                     , 

   restoId( NovoCar_1, ListaCar, NovoCar_2 ) ,   

   name( Id, [ Car | ListaCar ] )            , 

   ( serReserv( Id )                         ,

     Token = Id                              ; 

     Token = pred( Id ) )                    .

pegarToken( Car, var( Id ), NovoCar_2 )      :-  % lê variável

   serMaiuscula( Car )                       ,

   pegarCar( NovoCar_1 )                     ,

   restoId( NovoCar_1, ListaCar, NovoCar_2 ) ,

   name( Id, [ Car | ListaCar ] )            .

pegarToken( Car, Token, NovoCar_2 )        :-  % lê branco

   serBranco( Car )                        , 

   pegarCar( NovoCar )                     , 

   pegarToken( NovoCar, Token, NovoCar_2 ) .

pegarToken( Car, _, _ )              :-  % aponta caractere inválido

   write( '\n\nCaractere Ilegal: ' ) , 

   put( Car )                        , 

   nl                                , 

   abort                             .

restoId( Car, [ Car | ListaCar ], NovoCar_2 ) :- 

   serCarVal( Car )                           , 

   pegarCar( NovoCar_1 )                      ,

   restoId( NovoCar_1, ListaCar, NovoCar_2 )  .

restoId( Car, [ ], Car ) .

fazerAnaSintatica( Tokens, Arvore )                    :-

   write( '\nInterpretando: Análise sintática...\n' )  ,

   programa( Arvore, Tokens, [ eop ] )                 ,

   write( '\nAnálise sintática feita com sucesso.\n' ) ,

   write( '\narvore sintática = ' )                    ,

   write( Arvore )                                     ,

   nl                                                  .

% < programa > ::= < objetivo >

programa( prog( [ ], Objetivo ) ) -->

   [ se ]                         ,

   objetivo( Objetivo )           ,

   [ ponto ]                      .

% < programa > ::= < sentenças > < objetivo >

programa( prog( Sentencas, Objetivo ) ) -->

   sentencas( Sentencas )               ,

   [ se ]                               ,  

   objetivo( Objetivo )                 ,

   [ ponto ]                            .

% < sentenças > ::= < sentença >

sentencas( [ Sentenca ] ) -->

   sentenca( Sentenca )   ,

   [ ponto ]              .

% < sentenças > ::= < sentença > < sentenças >

sentencas( [ Sentenca | Sentencas ] ) -->

   sentenca( Sentenca )               ,

   [ ponto ]                          ,

   sentencas( Sentencas )             .

% < sentença > ::= < predicado > "."

sentenca( sent( pred( Pred ), [ ] ) ) -->

   [ pred( Pred ) ]                   .

% < sentença > ::= < predicado > ":-" < resto > "."

sentenca( sent( pred( Pred ), Resto ) ) -->

   [ pred( Pred ), se ]                 ,

   resto( Resto )                       .

% < objetivo > ::= ":-" < resto > "."

objetivo( obj( Objetivo ) ) -->

   resto( Objetivo )        .

% < resto > ::= < unidade >

resto( [ Unidade ] )  -->

   unidade( Unidade ) .

% < resto > ::= < unidade > "," < resto >

resto( [ Unidade | Resto ] ) -->

   unidade( Unidade )        ,

   [ virgula ]               ,

   resto( Resto )            .

% < unidade > ::= < predicado >

unidade( pred( Pred ) ) -->

   [ pred( Pred ) ]     .

% < unidade > ::= "fail"

unidade( fail ) -->

   [ fail ]     .

% < unidade > ::= "!"

unidade( corte ) -->

   [ corte ]     .

% < unidade > ::= "read(" < variável > ")"

unidade( read( var( Var ) ) )                   -->

   [ pred( read ), parene, var( Var ), parend ] .

% < unidade > ::= "write(" < variável > ")"

unidade( write( var( Var ) ) )                   --> 

   [ pred( write ), parene, var( Var ), parend ] .

fazerSOS( prog( ListaCla, obj( Obj ) ) )                              :-

   write( '\nInterpretando: SOS...\n' )                               ,

   write( '\nEntre lista de entrada entre [ ] e seguida de ponto: ' ) ,

   read( Ent )                                                        ,

   nl                                                                 ,

   mudar( Obj, NovoObj, 0, End )                                      ,

   imprimirObj( cfg( NovoObj, _ ) )                                   ,

   interpretar( cfg( NovoObj, est( Ent, [ ], End, [ ] ) )             ,

                est( _, Sai, _, _ ), ListaCla )                       ,

   write( '\nSaída = ' )                                              ,

   reverse( Sai, SaiInv )                                             ,   

   write( SaiInv )                                                    ,

   nl                                                                 .

mudar( Obj, NovoObj, EndAnt, EndNovo )     :-

   acharVars( Obj, [ ], Vars )             ,

   mudarVars( Obj, NovoObj, Vars, EndAnt ) ,

   length( Vars, NumVars )                 ,

   EndNovo is EndAnt + NumVars             .

acharVars( [ ], Vars, Vars ) .

acharVars( Obj, AntVars, NovaVars )                       :-

   ( Obj = [ read( var( Var ) ) | RestoObj ]              ;

     Obj = [ write( var( Var ) ) | RestoObj ] )           ,

   ( member( Var, AntVars )                               ->

     acharVars( RestoObj, AntVars, NovaVars )             ;

     acharVars( RestoObj, [ Var | AntVars ], NovaVars ) ) ,

   !                                                      .

acharVars( [ _ | RestoObj ], AntVars, NovaVars ) :-

   acharVars( RestoObj, AntVars, NovaVars )      .

mudarVars( [ ], [ ], _, _ ) .

mudarVars( Obj, NovoObj, Vars, EndAnt )              :-

   ( Obj = [ read( var( Var ) ) | RestoObj ]         ;

     Obj = [ write( var( Var ) ) | RestoObj ] )      ,

   nth0( Posicao, Vars, Var )                        ,

   End is EndAnt + Posicao                           ,

   atom_chars( End, ListaEnd )                       ,

   append( [ 'm' ], ListaEnd, ListaMem )             ,

   atom_chars( Mem, ListaMem )                       ,

   mudarVars( RestoObj, NovoResto, Vars, EndAnt )    ,

   ( Obj = [ read( var( Var ) ) | RestoObj ]         ->

     NovoObj = [ read( end( Mem ) ) | NovoResto ]    ;

     NovoObj = [ write( end( Mem ) ) | NovoResto ] ) ,

   !                                                 .

mudarVars( [ PrimObj | RestoObj ], [ PrimObj | NovoResto ], Vars, EndAnt ) :-

   mudarVars( RestoObj, NovoResto, Vars, EndAnt )                          .

cortar( Amb, NovoAmb, RaboObj, ListaCla )           :-

   cortarFilhos( Amb, TempAmb, RaboObj )            ,

   cortarPai( TempAmb, NovoAmb, RaboObj, ListaCla ) .

cortarFilhos( [ TopoAmb | RestoAmb ], NovoAmb, RaboObj ) :-

   append( _, RaboObj, TopoAmb )                         ,

   cortarFilhos( RestoAmb, NovoAmb, RaboObj )            .

cortarFilhos( Amb, Amb, _ ) .

cortarPai( [ TopoAmb | RestoAmb ], NovoAmb, RaboObj, ListaCla ) :-

   acharPai( TopoAmb, ListaCla, sent( Pai, _ ), Rabo )          ,

   append( RestoCla, Rabo, RaboObj )                            ,

   acharClausulas( ListaCla, Pai, Clausulas )                   ,

   acharCortaveis( Clausulas, RestoCla, Cortaveis )             ,

   append( Cortaveis, NovoAmb, [ TopoAmb | RestoAmb ] )         .

cortarPai( Amb, Amb, _, _ ) .

acharPai( TopoAmb, [ sent( Pai, Resto ) | _ ], sent( Pai, Resto ), Rabo ) :-

   append( Resto, Rabo, TopoAmb )                                         .

acharPai( TopoAmb, [ _ | RestoLista ], Sentenca, Rabo ) :-

   acharPai( TopoAmb, RestoLista, Sentenca, Rabo )      .

acharClausulas( [ ], _, [ ] ) .

acharClausulas( [ sent( Pai, Resto ) | Lista ], Pai, [ Resto | Clausulas ] ) :-

   acharClausulas( Lista, Pai, Clausulas )                                   .

acharClausulas( [ _ | Lista ], Pai, Clausulas ) :-

   acharClausulas( Lista, Pai, Clausulas )      .

acharCortaveis( [ Clausula | Cortaveis ], RestoCla, Cortaveis ) :-

   append( _, RestoCla, Clausula )                              .

acharCortaveis( [ _ | ListaCla ], RestoCla, Cortaveis ) :-

   acharCortaveis( ListaCla, RestoCla, Cortaveis )      .

modificar( [ ], [ ], _, _ ) .

modificar( Obj, NovoObj, Mem, Val )                  :-

   ( Obj = [ read( end( Mem ) ) | RestoObj ]         ;

     Obj = [ write( end( Mem ) ) | RestoObj ] )      ,

   modificar( RestoObj, NovoResto, Mem, Val )        ,

   ( Obj =  [ read( end( Mem ) ) | RestoObj ]        ->

     NovoObj = [ read( val( Val ) ) | NovoResto ]    ;

     NovoObj = [ write( val( Val ) ) | NovoResto ] ) .   

modificar( [ PrimObj | RestoObj ], [ PrimObj | NovoResto ], Mem, Val ) :-

   modificar( RestoObj, NovoResto, Mem, Val )                          . 

concatenar( [ ], _, [ ] ) .

concatenar( [ Cab | Resto ], Lista, [ CabRes | RestoRes ] ) :-

   append( Cab, Lista, CabRes )                             ,

   concatenar( Resto, Lista, RestoRes )                     .

imprimirObj( cfg( Obj, _)  ) :-

   write( 'Objetivo = ' )    ,

   write( Obj )              ,

   nl                        .

interpretar( cfg( [ ], EstFinal ), EstFinal, _ ) :-

   write( '\nYes' )                              .

interpretar( cfg( [ fail ], est( Ent, Sai, End, [ ] ) ), 

                            est( Ent, Sai, End, [ ] ), _ ) :-

   write( '\nNo' )                                         .

interpretar( Config, EstFinal, ListaCla )        :-

   transformar( Config, NovaConfig, ListaCla )   ,

   imprimirObj( NovaConfig )                     ,

   interpretar( NovaConfig, EstFinal, ListaCla ) .

% < :- true ., LC, E, S, M, A > -> < :- true ., LC, E, S, M, A >

transformar( cfg( [ ], EstFinal ), cfg( [ ], EstFinal ), _ ) .

% < :- fail ., LC, E, S, M, [ ] > -> < :- fail ., LC, E, S, M, [ ] >

% < :- fail, R ., LC, E, S, M, [ ] > -> < :- fail ., LC, E, S, M, [ ] >

transformar( cfg( [ fail | _ ], est( Ent, Sai, End, [ ] ) ),

             cfg( [ fail ], est( Ent, Sai, End, [ ] ) ), _ ) .

% < :- fail ., LC, E, S, M, [ TA | RA ] > -> < :- TA' ., LC, E, S, M', RA >

%           onde TA' = mud( TA, M ) e M' = M + vars( TA' )

% < :- fail, R ., LC, E, S, M, [ TA | RA ] > -> < :- TA' ., LC, E, S, M', RA >

%           onde TA' = mud( TA, M ) e M' = M + vars( TA' )

transformar( cfg( [ fail | _ ], est( Ent, Sai, End, [ TopoAmb | RestoAmb ] ) ),

             cfg( NovoObj, est( Ent, Sai, NovoEnd, RestoAmb ) ), _ ) :-

   mudar( TopoAmb, NovoObj, End, NovoEnd )                           .

% < :- ! ., LC, E, S, M, A > ->

%          < :- true ., LC, E, S, M, cortar( A, LC, true ) >

% < :- !, R ., LC, E, S, M, A > ->

%          < :- R ., LC, E, S, M, cortar( A, LC, R ) >

transformar( cfg( [ corte | RestoObj ], est( Ent, Sai, End, Amb ) ),

             cfg( RestoObj, est( Ent, Sai, End, NovoAmb ) ), ListaCla ) :-

   cortar( Amb, NovoAmb, [ corte | RestoObj ], ListaCla )               .

% < :- p ., LC, E, S, M, A > ->

%           < :- LC_1( p, true )' ., LC, E, S, M', [ LC_R( p, true ) | A ] >

%           onde LC_1( p, true )' = mud( LC_1( p, true ), M ) e

%                M' = M + vars( LC_1( p, true )' )

% < :- p, R ., LC, E, S, M, A > ->

%           < :- LCI_1( p, R )' ., LC, E, S, M', [ LC_R( p, R ) | A ] >

%           onde LC_1( p, R )' = mud( LC_1( p, R ), M ) e

%                M' = M + vars( LC_1( p, R )' )

transformar( cfg( [ pred( Pred ) | RestoObj ], est( Ent, Sai, End, Amb ) ),

             cfg( NovoObj, est( Ent, Sai, NovoEnd, NovoAmb ) ), ListaCla ) :-

   acharClausulas( ListaCla, pred( Pred ), [ Objetivo | Alternativas ] )   ,

   mudar( Objetivo, PrimObj, End, NovoEnd )                                ,

   append( PrimObj, RestoObj, NovoObj )                                    ,

   concatenar( Alternativas, RestoObj, Escolhas )                          ,

   append( Escolhas, Amb, NovoAmb )                                        .

% erro de predicado não definido no programa

transformar( cfg( [ pred( Pred ) | _ ], _ ), _, ListaCla ) :-

   acharClausulas( ListaCla, Pred, [ ] )                   ,

   write( '\nPredicado ' )                                 ,

   write( Pred )                                           ,

   write( ' não é definido.\n' )                           ,

   abort                                                   .

% lista de entrada vazia quando ocorre um read

transformar( cfg( [ read( _ ) | _ ], est( [ ], _, _, _ ) ), _, _ ) :-

   write( '\nErro: Não foi fornecida entrada.\n' )                 ,

   abort                                                           .

% < :- read( X ) ., LC, [ TE | RE ], S, M, A > ->

%           < :- true ., LC, RE, S, M, A >

% < :- read( X ), R ., LC, [ TE | RE ], S, M, A > ->

%           < :- R' ., LC, RE, S, M, A >

%           onde R' = trans( R, X, VE )

transformar( cfg( [ read( end( Mem ) ) | RestoObj ],

                  est( [ Val | Ent ], Sai, End, Amb ) ),

             cfg( NovoObj, est( Ent, Sai, End, Amb ) ), _ ) :-

   modificar( RestoObj, NovoObj, Mem, Val )                 .

% < :- read( V ) ., LC, [ V | RE ], S, M, A > -> < :- true ., LC, RE, S, M, A >

% < :- read( V ), R ., LC, [ V | RE ], S, M, A > -> < :- R ., LC, RE, S, M, A >

transformar( cfg( [ read( val( Val ) ) | RestoObj ],

                  est( [ Val | Ent ], Sai, End, Amb ) ),

             cfg( RestoObj, est( Ent, Sai, End, Amb ) ), _ ) .

% < :- read( V ) ., LC, [ TE | RE ], S, M, A > -> < :- fail ., LC, RE, S, M, A >

%          onde V ¹ TE

% < :- read( V ), R ., LC, [ TE | RE ], S, M, A > -> < :- fail ., LC, RE, S, M, A >

%          onde V ¹ TE

transformar( cfg( [ read( val( _ ) ) | _ ], est( [ _ | Ent ], Sai, End, Amb ) ),

             cfg( [ fail ], est( Ent, Sai, End, Amb ) ), _ ) . 

% < :- write( X ) ., LC, E, S, M, A > -> < :- true ., LC, E, [ _ | S ], M, A >

% < :- write( X ), p ., LC, E, S, M, A > -> < :- p ., LC, E, [ _ | S ], M, A >

transformar( cfg( [ write( end( _ ) ) | RestoObj ], est( Ent, Sai, End, Amb ) ),

             cfg( RestoObj, est( Ent, [ '_' | Sai ], End, Amb ) ), _ ) .

% < :- write( V ) ., LC, E, S, M, A > -> < :- true ., LC, E, [ V | S ], M, A >

% < :- write( V ), p ., LC, E, S, M, A > -> < :- p ., LC, E, [ V | S ], M, A >

transformar( cfg( [ write( val( Val ) ) | RestoObj ], est( Ent, Sai, End, Amb ) ),

             cfg( RestoObj, est( Ent, [ Val | Sai ], End, Amb ) ), _ ) .

3) SOS DE GOTO:


A linguagem usada para descrever a SOS do comando goto será Wren, a qual é definida em [1], extendida para suportar o uso deste comando. Para isso ser possível, primeiro é necessário extender a sintaxe abstrata, que é acrescida da seguinte regra:

Command ::= goto Label | Label : Command 


Feito isso, as regras de SOS incluídas são as seguintes:

< Label : Command, sto > ( < Command, sto[ Label/Command ] >

< goto Label, sto > ( < Command, sto >, onde 

sto( Label ) = Command

Na primeira regra, os comandos são armazenados na memória na posição apontada por Label, da mesma forma que se armazena inteiros em variáveis, com Command fazendo o papel de um inteiro e Label o de uma variável inteira. Já na segunda regra, quando o commando goto for ser executado, os comandos que passarão a ser executados serão os que seguem o rótulo dado no goto e que estão armazenados na memória na posição apontada pelo mesmo.  

Também é necessário modificar a regra da concatenação dois comandos (ou seqüência de comandos). Isto deve ser feito porque não necessáriamente o segundo será executado, se o primeiro for um goto. A regra, então, fica da seguinte forma, sendo dividida em duas:

    < c1, sto > ( < c1’, sto’ >  
< c1; c2, sto > ( < c1’; c2, sto’ > c1 ( goto Label

< c1; c2, sto > ( < c1, sto > c1 = goto Label

Nas regras acima, é importante observar que foram omitidas as listas de entrada e saída, já que não são necessárias para o goto.


Por fim, estas regras podem facilmente ser extendidas para linguagens com aninhamento de blocos, procedimentos e funções, como Pelican, que também é descrita em [1], levando em conta, além da memória, o ambiente.    

